```
 1: %% Implement merge sort.
 2:
 3: split(Whole, Left, Right) :-
 4:         split(Whole, Whole, Left, Right).
 5:
 6: split([], Right, [], Right).
 7: split([_], [X|Right], [X], Right). %% lean to the left
 8: split([_,_|Walker], [X|Whole], [X|Left], Right) :-
 9:         split(Walker, Whole, Left, Right).
10:
11: /*
12: merge(As,[],As,_Pred) :- !.
13: merge([],Bs,Bs,_Pred).
14: merge([A|As],[B|Bs],Cs,Pred) :-
15:         functor(X,Pred,2), arg(2,X,A), arg(1,X,B),
16:         call(X), !, Cs=[B|C1s], merge([A|As],Bs,C1s,Pred).
17: merge([A|As],Bs,[A|Cs],Pred) :- merge(As,Bs,Cs,Pred).
18: */
19:
20: merge([],Bs,Bs,_Pred).
21: merge([A|As],Bs,Cs,Pred) :-
22:         mergeR(Bs,A,As,Cs,Pred).
23:
24: mergeR([], A,As, [A|As],_).
25: mergeR([B|Bs],A,As,[B|Cs],Pred) :-
26:         functor(PredR,Pred,2),arg(2,PredR,A),arg(1,PredR,B),
27:         call(PredR), !, mergeR(Bs,A,As,Cs,Pred).
28: mergeR([B|Bs],A,As,[A|Cs],Pred) :-
29:         !, mergeL(As,B,Bs,Cs,Pred).
30:
31: mergeL([],B,Bs,[B|Bs],_).
32: mergeL([A|As],B,Bs,[A|Cs],Pred) :-
33:         functor(PredL,Pred,2), arg(1,PredL,B),
34:         arg(2,PredL,A), \+(PredL), !,
35:         mergeL(As,B,Bs,Cs,Pred).
36: mergeL([A|As],B,Bs,[B|Cs],Pred) :-
37:         !, mergeR(Bs,A,As,Cs,Pred).
38:
39:
40: mergeSort([],[],_).
41: mergeSort([A],[A],_).
42: mergeSort(Ins,Outs,Pred) :-
43:         split(Ins,Lefts,Rights),
44:         mergeSort(Lefts,OLefts,Pred),
45:         mergeSort(Rights,ORights,Pred),
46:         merge(OLefts,ORights,Outs,Pred).
47:
48:
49: randSplit(Xs,Lefts,Rights) :-
50:         length(Xs,N), N2 is (N+1)//2,R2 is N-N2,
51:         randSplit(N2,R2,Xs,Lefts,Rights).
52: randSplit(0,LR,Ins,[],Ins) :- !, length(Ins,LR).
53: randSplit(LL,0,Ins,Ins,[]) :- !, length(Ins,LL).
54: randSplit(N2,R2,Xs,Lefts,Rights) :-
55:         random(N2+R2)<N2, !, succ(N2P,N2), Xs=[X|XsP], Lefts=[X|LeftsP],
56:         randSplit(N2P,R2,XsP,LeftsP,Rights).
57: randSplit(N2,R2,Xs,Lefts,Rights) :-
58:         succ(R2P,R2), Xs=[X|XsP], Rights=[X|RightsP],
59:         randSplit(N2,R2P,XsP,Lefts,RightsP).
60: randShuffle([],[]) :- !.
61: randShuffle([X],[X]) :- !.
62: randShuffle(Xs,Ys) :-
63:         randSplit(Xs,Zs,Ws),
64:         randShuffle(Zs,Zs1),
65:         randShuffle(Ws,Ws1),
66:         append(Zs1,Ws1,Ys).
67:
68: randList(K,List) :-
69:         numlist(1,K,L1), randShuffle(L1,List).
```